Renormalization Group

We have learned high and low Temperature expansions, transfer matrix, and mean field theory. Another powerful technique for statistical mechanics of interacting systems and their phase transitions is the renormalization group.

The basic idea is that if there is a diverging correlation length ξ all smaller length scales ℓ become irrelevant since $\ell/\xi \to \phi$. A profound consequence is that short length scale details of the model do not affect certain "universal" features of the physics, specifically the critical exponents are unchanged if we make changes like $E = -J \sum_{\langle ij \rangle} S_i S_j \to$

$$E = -J \sum_{\langle ij \rangle} S_i S_j - K \sum_{\langle\langle ik \rangle\rangle} S_i S_k$$

near neighbor

next near neighbor



RG-2

ν, δ, β etc are the same for all values of K!

This makes us a lot more comfortable with using models! Their details do not matter. They make accurate predictives for real materials which have different details of interactions.

Note: T_c is <u>not</u> universal.

Note: Exponents do depend on dimensionality of lattice and symmetry of order parameter ("n" and "d" in previous notes). Thus Ising and XY models have different exponents in $d=2$, and Ising has different exponents in $d=2$ and $d=3$.

A more refined picture is in terms of RG "flows".



as length changes K, J change. They "flow" to or from a "fixed point" and all phys
is determined there. So all
attracted to same fixed
same exponent

RG 3

To do RG then you need to figure out how $J, K, \ldots$ change with length scale.

Let's do a very simple example which reproduces our result that $T_c = 0$ for the $d=1$ Ising model



change length scale by factor of 2

We need

$$e^{\beta J' S_1 S_3} = c \sum_{S_2} e^{\beta J (S_1 S_2 + S_2 S_3)}$$

This will guarantee $Z' = Z$,

S_1	S_3	$e^{\beta J' S_1 S_3}$	$c \sum_{S_2} e^{\beta J (S_1 S_2 + S_2 S_3)}$
$+$	$+$	$e^{\beta J'}$	$c(e^{2\beta J} + e^{-2\beta J})$
$+$	$-$	$e^{-\beta J'}$	$c(1+1)$

other two choices provide no new info

Thus $$e^{\beta J'} = \frac{1}{2} e^{-\beta J'} 2\cosh 2\beta J$$

$$e^{2\beta J'} = \cosh 2\beta J \Rightarrow \beta J' = \frac{1}{2} \ln(\cosh 2\beta J)$$

$$\beta J' = \frac{1}{2} \ln(e^{2\beta J} + e^{-2\beta J})/2 = \frac{1}{2} \ln e^{2\beta J}(1 + e^{-4\beta J})/2$$

$$= \underbrace{\frac{1}{2} 2\beta J}_{\beta J} \underbrace{- \frac{1}{2} \ln 2 + \frac{1}{2} \underbrace{\ln(1 + e^{-4\beta J})}_{\text{this is } < 2}}_{\text{so this is } < 0}$$

So $\beta J' < \beta J$ we flow to $\beta J = 0$ ie to $T = \infty$, the disordered phase.

$$\beta J' = \beta J - \tfrac{1}{2}\ln 2 + \tfrac{1}{2}\ln\left(1+e^{-4\beta J}\right)$$

$\left.\begin{array}{l} \beta J' = \beta J = 0 \\ \beta J' = \beta J = \infty \end{array}\right\}$ are "fixed point". They are sol'ns of RG flow eqn

RG flows for $d=1$ Ising model:



$(\beta J = \infty)$ is low T fixed point. It is unstable

$(\beta J = 0)$ is high T fixed point. It is stable

$d=1$ Ising model flows to $T=\infty$ $(\beta = 0)$ fixed point for <u>any</u> $\beta J > 0$ $\Rightarrow$ no phase transitions.

In $d=2$ we would get



Comments: ① There are different techniques for changing length scale and generating the RG flow eqn. Except in a very few simple eqn they are <u>approximate</u>. So RG does not in general solve problems exactly.

② This scheme of integrating out sites is sometimes called "decimation". Does anyone know origin of word "decimation".

③ RG is often phrased (esp in HE) as changing <u>energy</u> <u>scale</u>. This is same as changing length scale because integrating out short lengths is like integrating out high <u>momentum</u>, ie high <u>energy</u>.

RG 5

An example of an approximate RG analysis based on "block spins": The 2D Ising model on a triangular lattice



We use the "majority rule"

$$S_1' = \begin{cases} +1 & S_1+S_2+S_3>0 \\ -1 & S_1+S_2+S_3<0 \end{cases}$$ and similarly for S_2'



$$E = -J(S_1S_2+S_1S_3+S_2S_3) - J(S_2S_4+S_3S_4) - J(S_4S_5+S_4S_6+S_5S_6) - B(S_1+S_2+S_3+S_4+S_5+S_6)$$

$$E' = -J'(S_1'S_2') - B'(S_1'+S_2')$$

S_1	S_2	S_3	S_4	S_5	S_6	S_1'	S_2'	E	E'
+	+	+	+	+	+	+	+	$-8J-6B$	$-J'-2B'$
+	+	+	+	+	−	+	+		$-J'-2B'$
		⋮							
+	+	+	−	−	−	+	−	$-4J$	$+J'$

There are 64 choices (2^6) of $S_1\,S_2\,S_3\,S_4\,S_5\,S_6$.
16 of these give rise to $S_1'=+$ $S_2'=+$; 16 to $S_1'=+$ $S_2'=-$, etc

Write a little code which computes

$$c\,e^{\beta J'+2\beta B'} = \sum_{\substack{16\text{ choices}\\ \text{for } S_1'S_2'=++}} e^{-\beta E} \equiv Z_{++}$$

$$c\,e^{-\beta J'} = \sum_{\substack{16\text{ choices}\\ \text{for } S_1'S_2'=+-}} e^{-\beta E} \equiv Z_{+-}$$

get no new information from $S_1'S_2' = -+$

$$c\,e^{\beta J'-2\beta B'} = \sum_{16\text{ choices } S_1'S_2'=--} e^{-\beta E} \equiv Z_{--}$$

RG6

These give 3 eqns in 3 unknowns C, J', B'

$$Z_{++}(Z_{+-})^2 Z_{--} = C^4 \quad \text{determines } C$$

$$Z_{++}/Z_{--} = e^{4\beta B'} \quad \text{determines } B'$$

$$Z_{+-}/C = e^{-\beta J'} \quad \text{determines } J' \text{ since}$$

C is known

Given initial J, B can get new J' B' (numerically)
This gives RG flow, we expect fixed point at
$B=0$ and nontrivial βJ (ie $\beta J \neq 0, \infty$).

Comment: We can see some things even without program.
For example, at $B=0$ it is clear $Z_{++} = Z_{--}$
Looking at middle eqn $Z_{++}/Z_{--} = e^{4\beta B'}$
we see $B'=0$ if $B=0$. So flows on J-axis
stay on J-axis.

Fixed points play crucial role since
correlation length $\xi = \infty$ there, so they correspond
to critical points. To see this, note the
meaning of fixed point is that $J, B, K, \ldots$ unchanged
when the length scale changes. So physics there
is scale invariant. The only way this could happen
is if $\xi = \infty$.

RG 7



R48

```
c       CODE TO GET RG FLOWS USING MAJORITY RULE FOR 2D TRI LATTICE

        implicit none
        real*8 J,B,C,Jp,Bp,zpp,zpm,zmm,E
        integer S1,S2,S3,S4,S5,S6,it,Nit

        write (6,*) 'enter initial J,B,#iterations'
        read  (5,*)                J,B,Nit
        write (6,*) '   it       C            B           J  '

        do 1000 it=1,Nit

             zpp=0.d0
             zpm=0.d0
             zmm=0.d0

             do 60 S1=-1,1,2
             do 50 S2=-1,1,2
             do 40 S3=-1,1,2
             do 30 S4=-1,1,2
             do 20 S5=-1,1,2
             do 10 S6=-1,1,2

                  E=  J*(S1*S2+S1*S3+S2*S3)+B*(S1+S2+S3)
                  E=E+J*(S4*S5+S4*S6+S5*S6)+B*(S4+S5+S6)
                  E=E+J*(S4*S2+S4*S3)

                  if (S1+S2+S3.gt.0.and.S4+S5+S6.gt.0) then
                      zpp=zpp+dexp(E)
                  endif

                  if (S1+S2+S3.gt.0.and.S4+S5+S6.lt.0) then
                      zpm=zpm+dexp(E)
                  endif

                  if (S1+S2+S3.lt.0.and.S4+S5+S6.lt.0) then
                      zmm=zmm+dexp(E)
                  endif

10           continue
20           continue
30           continue
40           continue
50           continue
60           continue

             C  = dsqrt( dsqrt(zpp*zpm*zpm*zmm) )
             Bp = 0.25d0*dlog(zpp/zmm)
             Jp = -dlog(zpm/C)

             write  (6,991) it,C,Bp,Jp
             write (66,991) it,C,Bp,Jp
991          format(i6,e12.4,2f12.6)

             B=Bp
             J=Jp

1000    continue

        end
```

RG9A

More formal presentation of block spin procedure

Require
$$Z = \sum_{\{S\}}' e^{-H(S)} = \sum_{\{S'\}} e^{-H'(S')}$$

(we have included β factor in $\mathcal{H}$)

Define a projection operator eg

$$T_b(S'|S) \equiv \begin{cases} 1 & S'_\ell \left(\sum'_{\ell \in \text{Block}} S_\ell \right) > 0 \\ \phi & S'_\ell \left(\sum_{\ell \in \text{Block}} S_\ell \right) < \phi \end{cases}$$

Can guarantee $*$ is true if

$$e^{-H'(S')} = \sum_{\{S\}}' \prod_{\text{Blocks}} T_b(S,S') e^{-H(S)}$$

This defines $H'(S')$

RG 9B

Change length scale $\ell \to b\,\ell$

Parameters in Hamiltonian change $J \to J'$ really βJ

$B \to B'$ βB

$K \to K'$ βK

Correlation length ξ is a function of eg in

$d=1$ Ising model $\quad \xi = -1/\ln(\tanh \beta J)$

If $\left.\begin{array}{l} J' = J \\ B' = B \\ K' = K \\ \dots \end{array}\right\}$ "fixed point" $\quad \xi$ unchanged because $\xi \to \xi/b$

$\Rightarrow \xi = 0, \infty$

critical point

$d=1$ Ising $\quad \bullet \longrightarrow \bullet$

$\beta J = 0$ high T $\beta J = \infty$ low T

$d=2$ Ising (Δ lattice $\quad \bullet \text{---} \bullet \longrightarrow \bullet$

$\beta J = 0$ $\beta J = 0.365$ $\beta J = \infty$

RG-10

Let's use $\tilde{J}$ for βJ and $\tilde{J}_*$ for the fixed point / critical point. $\xi(T) \sim (T-T_c)^{-\nu}$ can be written as $\xi(\tilde{J}) = A/(\tilde{J}-\tilde{J}_*)^{\nu}$ in this notation.

If the length scale $l \to bl$ the correlation length measured in units of lattice spacing is decreased by a factor of l so $\xi(\tilde{J}') = b\,\xi(\tilde{J})$

$$A(\tilde{J}-\tilde{J}_*)^{-\nu} = bA(\tilde{J}'-\tilde{J}_*)^{-\nu}$$

$$\tilde{J}'-\tilde{J}_* = \underbrace{\left.\frac{\partial \tilde{J}'}{\partial \tilde{J}}\right|_{\tilde{J}=\tilde{J}_*}}_{\text{define this as } \lambda} (\tilde{J}-\tilde{J}_*)$$

(This is just Taylor expansion of $\tilde{J}'(\tilde{J})$ around $\tilde{J}=\tilde{J}_*$)

$$1 = b\lambda^{-\nu} \qquad \nu = \ln b / \ln \lambda$$

Words: To compute critical exponent ν, look at slope of $\tilde{J}'(\tilde{J})$ near critical point in terms of how much the length scale has changed.

Our problem for Δ lattice blocking



$b = \sqrt{3}$ ← why?

well 3 spins coalesce into 1 spin so any area A containing N spins now contains N/3 spins. so their linear separation must increase by $\sqrt{3}$

If we explicitly write out eqns on page 5 we get

(taking ratio of eqns for $(S_1', S_2') = (+,+)$ to $(S_1', S_2') = (-,-)$

$$e^{2\tilde{J}'} = \frac{e^{8\tilde{J}} + e^{2\tilde{J}}\left(3e^{2\tilde{J}} + e^{-2\tilde{J}} + 2\right) + e^{-2\tilde{J}}\left(2e^{2\tilde{J}} + e^{-2\tilde{J}} + 6\right)}{e^{4\tilde{J}} + e^{2\tilde{J}}\left(e^{2\tilde{J}} + 3e^{-2\tilde{J}} + 2\right) + e^{-2\tilde{J}}\left(e^{2\tilde{J}} + 2e^{-2\tilde{J}} + 6\right)}$$

$X \equiv e^{2\tilde{J}}$

$$X' = \frac{x^4 + 3x^2 + 2x + 3 + 6/x + 1/x^2}{2x^2 + 2x + 4 + 6/x + 2/x^2}$$

$$\lambda = \left.\frac{\partial \tilde{J}'}{\partial \tilde{J}}\right|_{\tilde{J} = \tilde{J}_*} = \left.\frac{\partial x'}{\partial x}\right|_{X = x_*}$$

because $\frac{\partial \tilde{J}'}{\partial \tilde{J}} = \frac{\partial \tilde{J}'}{\partial x'}\frac{\partial x'}{\partial x}\frac{\partial X}{\partial \tilde{J}} = \frac{1}{2x'}\frac{\partial x'}{\partial x} 2x$ and $\frac{X}{x'} = 1$ at x_*

Need to get $\partial x'/\partial x$ at $x_* = e^{2(0.365)}$ $\tilde{J}_*$

It turns out to be 1.54

$$\nu = \ln\sqrt{3}/\ln(1.54) = 1.26$$

Exact ν for 2D Ising is $\nu = 1$.

Summary:

(1) Change scale (decimation, block spins) of length

(2) compute $\tilde{J}'(\tilde{J})$

(3) fixed points $\Rightarrow T_c$

(4) $\partial\tilde{J}'/\partial\tilde{J} \Rightarrow \nu$.

can also get other exponents

RG 13

```
c        CODE TO GET dx'/dx SO TO OBTAIN lambda
c        RESULTS:
c        enter dx
c       .001
c        xp1=x?   2.0763133539232017        2.0763135288512800
c        dxp/dx (Cardy says 1.54)   1.5440012359762001

         implicit none
         real*8 x,xp1,dx,xp2,deriv,Kcrit

         write (6,*) 'enter dx'
         read  (5,*)        dx

c        THIS Kcrit VALUE FROM rgtri1.f:
         Kcrit=0.365297
         x=dexp(2.d0*Kcrit)

         xp1 = (x**4 + 3.d0*x**2 + 2.d0*x + 3.d0 + 6.d0/x + 1.d0/x**2)
         xp1 = xp1 /
     1       (2.d0*x**2 + 2.d0*x + 4.d0 + 6.d0/x + 2.d0/x**2)

         write (6,*) 'xp1=x?', xp1,x

         x=x+dx
         xp2 = (x**4 + 3.d0*x**2 + 2.d0*x + 3.d0 + 6.d0/x + 1.d0/x**2)
         xp2 = xp2 /
     1       (2.d0*x**2 + 2.d0*x + 4.d0 + 6.d0/x + 2.d0/x**2)

         deriv = (xp2-xp1)/dx

         write (6,*) 'dxp/dx (Cardy says 1.54)',deriv

         end
```